
1 Overview

This document lists all the changes needed to correct issues in the ATS iWD-TTCN3-B2017-06_D18wk12 related to the title of this CR.

Contact:
Hans Rohnert

hans.rohnert@rohde-schwarz.com
2 Corrections required

2.1 f_IMS_MTSpeechAddVideoRemoveVideo
	Function name
	f_IMS_MTSpeechAddVideoRemoveVideo

	Reason for change
	TS 34.229-1, Step 6 of IMS test case 17.2 states in Note 1 that, for the video media stream attributes, the strength value for the UE leg has to be taken from what the UE had delivered in 183 Session Progress. However, TTCN simply takes the strength value from the first set of attributes, potentially happening to be the attributes for the audio media stream.

	Summary of change
	Iterate over the media lines until the video media line, and take the strength tag from there.

Note: replacing media_list[0] by media_list[1] would likely be wrong as the UE might bring the video media first.

	TTCN module
	IMS_CC_TestcaseImplementation_17.ttcn

	MCC160 Comment
	


Before change
    function f_IMS_MTSpeechAddVideoRemoveVideo() runs on IMS_PTC

  {

    var IMS_CC_MT_SpeechCallInfo_Type v_SpeechCallInfo;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var IMS_Response_Type v_SessionInProgress183;

    var template (value) INVITE_Request v_InviteRequest;

    var template (value)   SDP_Message v_SdpOffer;

    var template (present) SDP_Message v_SdpAnswer;

    var template SDP_Message v_SdpAnswer200OK;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) MessageHeader v_MessageHeader_Update;

    var template (value) MessageHeader v_MessageHeader_Ack;

    var template (present) MessageHeader v_MessageHeader_Response;

    var template (omit) SDP_attribute v_SDP_Attribute;

    var SipUrl v_ContactUrl;

    var InternetProtocol_Type v_Protocol;

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var IMS_SDP_AudioCodecList_Type v_AudioCodecList;

    var boolean v_PreconditionsRequired;

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();

    if (v_IsFBBA or v_IsWLAN) {

      if (v_IsFBBA) {

        v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_AMR_only;

        v_PreconditionsRequired := false;

      } else {

        v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;

        v_PreconditionsRequired := true;

      }

      v_SpeechCallInfo := f_IMS_MTCallSetup_AnnexC11aC11c(v_AudioCodecList, v_PreconditionsRequired);

    }

    else {

      v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;

      v_SpeechCallInfo := f_IMS_MTCallSetup_AnnexC11C11b(v_AudioCodecList);

    }

    v_ContactUrl := v_SpeechCallInfo.ContactUrl;

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    // Step 1 SS sends re-INVITE with second SDP offer to add video.

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(AddVideo, Invite, v_ContactUrl);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    if (v_IsFBBA or v_IsWLAN) {

      // @siclog "Step 2" siclog@

      v_Protocol := v_InviteRequestWithSdp.RoutingInfo.Protocol;

      fl_IMS_ReceiveOptionalTrying_Timeout(v_InviteRequest, v_InviteRequestWithSdp.RoutingInfo.Protocol);

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MT_AddVideo);

      f_IMS_IPCAN_WaitForTrigger(IPCAN);

      v_SdpAnswer200OK := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo);

      // @siclog "Step 3 - 5: Void (H.17.2)" siclog@

    }

    else {

      // @siclog "Step 2: Void" siclog@

      // @siclog "Step 3" siclog@

      v_ResponseWithSdp := f_IMS_MTCallSetup_Video_Receive183(AddVideo, v_InviteRequestWithSdp);    /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_SessionInProgress183 := v_ResponseWithSdp.Response;

      // @siclog "Step 4 - 5" siclog@

      f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionInProgress183.msgHeader);

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MT_AddVideo);            /* @sic R5-181141: trigger moved after step 4/5 sic@ */

      f_IMS_IPCAN_WaitForTrigger(IPCAN);

      // @siclog "Step 6" siclog@

      v_SDP_Attribute := f_SDP_AttributeList_GetAttribute(v_ResponseWithSdp.SdpMessage.media_list[0].attributes, cr_SDP_Attribute_curr_qos(c_local, ?));
      v_SdpOffer := f_IMS_BuildSDP_MTCallAudioVideo_TX(AddVideo, Update, valueof(v_SDP_Attribute.curr.direction)); /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(v_InviteRequest, PreconditionSupported);     /* @sic R5-160795: PreconditionSupported sic@ */

      v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();                                                    /* @sic R5s160540 change 1.5 sic@ */

      IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_Protocol), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(v_SdpOffer))));

      // @siclog "Step 7" siclog@

      v_SdpAnswer := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo, Response200);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update, cr_ContentTypeSDP);   /* @sic R5-160795, R5-162930: precondition required sic@ */

      v_IMS_DATA_RSP := f_IMS_ReceiveResponse(v_MessageHeader_Update, v_Protocol, cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP));

      f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_DATA_RSP.Response.messageBody, v_SdpAnswer);

      v_SdpAnswer200OK := omit;

    }

    // @siclog "Step 7A (17.2)" siclog@

    // @siclog "Step 2A (G.17.2)" siclog@

    // @siclog "Step 6 (17.2)" siclog@

    f_UT_AnswerCall(MMI);

    // @siclog "Step 8 - 9 (17.2)" siclog@

    // @siclog "Step 3 - 4 (G.17.2)" siclog@

    // @siclog "Step 7 - 8 (H.17.2)" siclog@

    // The UE responds to the re-INVITE with a 200 OK final response.

    // The SS acknowledges the receipt of 200 OK for the re-INVITE.

    f_IMS_MTCallSetup_Common_Steps12_13(v_InviteRequestWithSdp, v_SdpAnswer200OK);

    // @siclog "Step 10 (17.2)" siclog@

    // @siclog "Step 5 (G.17.2)" siclog@

    // @siclog "Step 9 (H.17.2)" siclog@

    // SS sends a re-INVITE with a SDP offer indicating that the video component is removed from the call

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(RemoveVideo, Invite, v_ContactUrl);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

    // @siclog "Step 10A - 11 (17.2)" siclog@

    // @siclog "Step 6 - 7 (G.17.2)" siclog@

    // @siclog "Step 10 - 11 (H.17.2)" siclog@

    v_MessageHeader_Invite := v_InviteRequestWithSdp.Invite.msgHeader;

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Invite, cr_ContentTypeSDP);    /* @sic R5-160795, R5-162930: cr_ContentTypeSDP, precondition required sic@ */

    f_IMS_ReceiveResponse_OptionalTrying(v_InviteRequestWithSdp.Invite,

                                         v_InviteRequestWithSdp.RoutingInfo.Protocol,

                                         cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP));  /* @sic R5s160842, R5s170013 sic@ */

    // @siclog "Step 11A (17.2)" siclog@

    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_ReleaseVideo);

    // @siclog "Step 12 (17.2)" siclog@

    // @siclog "Step 8  (G.17.2)" siclog@

    // @siclog "Step 12 (H.17.2)" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite);

    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_InviteRequestWithSdp.RoutingInfo.Protocol), cs_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

    f_IMS_IPCAN_WaitForTrigger(IPCAN);    // Wait for trigger from IPCAN to ensure that radio resources have been released

    // @siclog "Step 13: Void (17.2)" siclog@

    // @siclog "Step 14 - 16 (17.2)" siclog@

    // @siclog "Step 9 - 10  (G.17.2)" siclog@

    // @siclog "Step 13 - 14 (H.17.2)" siclog@

    f_IMS_CallReleaseMT(v_InviteRequest, v_ContactUrl);

    f_IMS_TestBody_Set(false);

  }
After change
   function f_IMS_MTSpeechAddVideoRemoveVideo() runs on IMS_PTC

  {

    var IMS_CC_MT_SpeechCallInfo_Type v_SpeechCallInfo;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var IMS_Response_Type v_SessionInProgress183;

    var template (value) INVITE_Request v_InviteRequest;

    var template (value)   SDP_Message v_SdpOffer;

    var template (present) SDP_Message v_SdpAnswer;

    var template SDP_Message v_SdpAnswer200OK;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) MessageHeader v_MessageHeader_Update;

    var template (value) MessageHeader v_MessageHeader_Ack;

    var template (present) MessageHeader v_MessageHeader_Response;

    var template (omit) SDP_attribute v_SDP_Attribute;

    var SipUrl v_ContactUrl;

    var InternetProtocol_Type v_Protocol;

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var IMS_SDP_AudioCodecList_Type v_AudioCodecList;

    var boolean v_PreconditionsRequired;

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();

    var integer i;
    if (v_IsFBBA or v_IsWLAN) {

      if (v_IsFBBA) {

        v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_AMR_only;

        v_PreconditionsRequired := false;

      } else {

        v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;

        v_PreconditionsRequired := true;

      }

      v_SpeechCallInfo := f_IMS_MTCallSetup_AnnexC11aC11c(v_AudioCodecList, v_PreconditionsRequired);

    }

    else {

      v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;

      v_SpeechCallInfo := f_IMS_MTCallSetup_AnnexC11C11b(v_AudioCodecList);

    }

    v_ContactUrl := v_SpeechCallInfo.ContactUrl;

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    // Step 1 SS sends re-INVITE with second SDP offer to add video.

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(AddVideo, Invite, v_ContactUrl);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    if (v_IsFBBA or v_IsWLAN) {

      // @siclog "Step 2" siclog@

      v_Protocol := v_InviteRequestWithSdp.RoutingInfo.Protocol;

      fl_IMS_ReceiveOptionalTrying_Timeout(v_InviteRequest, v_InviteRequestWithSdp.RoutingInfo.Protocol);

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MT_AddVideo);

      f_IMS_IPCAN_WaitForTrigger(IPCAN);

      v_SdpAnswer200OK := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo);

      // @siclog "Step 3 - 5: Void (H.17.2)" siclog@

    }

    else {

      // @siclog "Step 2: Void" siclog@

      // @siclog "Step 3" siclog@

      v_ResponseWithSdp := f_IMS_MTCallSetup_Video_Receive183(AddVideo, v_InviteRequestWithSdp);    /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_SessionInProgress183 := v_ResponseWithSdp.Response;

      // @siclog "Step 4 - 5" siclog@

      f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionInProgress183.msgHeader);

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MT_AddVideo);            /* @sic R5-181141: trigger moved after step 4/5 sic@ */

      f_IMS_IPCAN_WaitForTrigger(IPCAN);

      // @siclog "Step 6" siclog@

      for (i:=0; i < lengthof(v_ResponseWithSdp.SdpMessage.media_list); i:=i+1) {

        if (v_ResponseWithSdp.SdpMessage.media_list[i].media_field.media == "video") {

          v_SDP_Attribute := f_SDP_AttributeList_GetAttribute(v_ResponseWithSdp.SdpMessage.media_list[i].attributes, cr_SDP_Attribute_curr_qos(c_local, ?));

        }

      }
      v_SdpOffer := f_IMS_BuildSDP_MTCallAudioVideo_TX(AddVideo, Update, valueof(v_SDP_Attribute.curr.direction)); /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(v_InviteRequest, PreconditionSupported);     /* @sic R5-160795: PreconditionSupported sic@ */

      v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();                                                    /* @sic R5s160540 change 1.5 sic@ */

      IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_Protocol), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(v_SdpOffer))));

      // @siclog "Step 7" siclog@

      v_SdpAnswer := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo, Response200);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update, cr_ContentTypeSDP);   /* @sic R5-160795, R5-162930: precondition required sic@ */

      v_IMS_DATA_RSP := f_IMS_ReceiveResponse(v_MessageHeader_Update, v_Protocol, cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP));

      f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_DATA_RSP.Response.messageBody, v_SdpAnswer);

      v_SdpAnswer200OK := omit;

    }

    // @siclog "Step 7A (17.2)" siclog@

    // @siclog "Step 2A (G.17.2)" siclog@

    // @siclog "Step 6 (17.2)" siclog@

    f_UT_AnswerCall(MMI);

    // @siclog "Step 8 - 9 (17.2)" siclog@

    // @siclog "Step 3 - 4 (G.17.2)" siclog@

    // @siclog "Step 7 - 8 (H.17.2)" siclog@

    // The UE responds to the re-INVITE with a 200 OK final response.

    // The SS acknowledges the receipt of 200 OK for the re-INVITE.

    f_IMS_MTCallSetup_Common_Steps12_13(v_InviteRequestWithSdp, v_SdpAnswer200OK);

    // @siclog "Step 10 (17.2)" siclog@

    // @siclog "Step 5 (G.17.2)" siclog@

    // @siclog "Step 9 (H.17.2)" siclog@

    // SS sends a re-INVITE with a SDP offer indicating that the video component is removed from the call

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(RemoveVideo, Invite, v_ContactUrl);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

    // @siclog "Step 10A - 11 (17.2)" siclog@

    // @siclog "Step 6 - 7 (G.17.2)" siclog@

    // @siclog "Step 10 - 11 (H.17.2)" siclog@

    v_MessageHeader_Invite := v_InviteRequestWithSdp.Invite.msgHeader;

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Invite, cr_ContentTypeSDP);    /* @sic R5-160795, R5-162930: cr_ContentTypeSDP, precondition required sic@ */

    f_IMS_ReceiveResponse_OptionalTrying(v_InviteRequestWithSdp.Invite,

                                         v_InviteRequestWithSdp.RoutingInfo.Protocol,

                                         cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP));  /* @sic R5s160842, R5s170013 sic@ */

    // @siclog "Step 11A (17.2)" siclog@

    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_ReleaseVideo);

    // @siclog "Step 12 (17.2)" siclog@

    // @siclog "Step 8  (G.17.2)" siclog@

    // @siclog "Step 12 (H.17.2)" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite);

    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_InviteRequestWithSdp.RoutingInfo.Protocol), cs_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

    f_IMS_IPCAN_WaitForTrigger(IPCAN);    // Wait for trigger from IPCAN to ensure that radio resources have been released

    // @siclog "Step 13: Void (17.2)" siclog@

    // @siclog "Step 14 - 16 (17.2)" siclog@

    // @siclog "Step 9 - 10  (G.17.2)" siclog@

    // @siclog "Step 13 - 14 (H.17.2)" siclog@

    f_IMS_CallReleaseMT(v_InviteRequest, v_ContactUrl);

    f_IMS_TestBody_Set(false);

  } 
